<commit_before># Configure paths for libogg
# updated by <NAME> 10-Jun-2003
# <NAME> <<EMAIL>> 10-21-2000
# Shamelessly stolen from <NAME> and <NAME>

dnl XIPH_PATH_OGG([ACTION-IF-FOUND [, ACTION-IF-NOT-FOUND]])
dnl Test for libogg, and define OGG_CFLAGS OGG_LDFLAGS and OGG_LIBS
dnl
AC_DEFUN(XIPH_PATH_OGG,
[dnl 
AC_ARG_VAR([OGG_PREFIX],[path to ogg installation])
AC_ARG_WITH(ogg,
    [AC_HELP_STRING([--with-ogg=PREFIX],
                   [Prefix where libogg is installed (optional)])],
    ogg_prefix="$withval",
    ogg_prefix="$OGG_PREFIX"
)
if test "x$ogg_prefix" = "x"; then
    if test "x$prefix" = "xNONE"; then
        ogg_prefix=/usr/local
    else
        ogg_prefix="$prefix"
    fi
fi

XIPH_GCC_WARNING([-I$ogg_prefix/include],,
        [OGG_CFLAGS="-I$ogg_prefix/include"
        OGG_LDFLAGS="-L$ogg_prefix/lib"
        ])
OGG_LIBS="-logg"

#
# check if the installed Ogg is sufficiently new.
#
AC_MSG_CHECKING([for ogg_sync_init in libogg])
ac_save_CFLAGS="$CFLAGS"
ac_save_LIBS="$LIBS"
ac_save_LDFLAGS="$LDFLAGS"
CFLAGS="$CFLAGS $OGG_CFLAGS"
LIBS="$LIBS $OGG_LIBS"
LDFLAGS="$LDFLAGS $OGG_LDFLAGS"
AC_TRY_LINK_FUNC(ogg_sync_init,
        [ifelse([$1],, [AC_MSG_RESULT([ok])], [$1])],
        [AC_TRY_LINK([#include <ogg/ogg.h>],, 
            [ ifelse([$2], ,[AC_MSG_ERROR([found, but needs updating])], [$2])],
            [ ifelse([$2], ,[AC_MSG_ERROR([not found, maybe you need to set LD_LIBRARY_PATH or /etc/ld.so.conf])], [$2])])
        ])
CFLAGS="$ac_save_CFLAGS"
LIBS="$ac_save_LIBS"

AC_SUBST(OGG_CFLAGS)
AC_SUBST(OGG_LDFLAGS)
AC_SUBST(OGG_LIBS)
])
<commit_msg>Reset LDFLAGS after the link test.<commit_after># Configure paths for libogg
# updated by <NAME> 10-Jun-2003
# <NAME> <<EMAIL>> 10-21-2000
# Shamelessly stolen from <NAME> and <NAME>

dnl XIPH_PATH_OGG([ACTION-IF-FOUND [, ACTION-IF-NOT-FOUND]])
dnl Test for libogg, and define OGG_CFLAGS OGG_LDFLAGS and OGG_LIBS
dnl
AC_DEFUN(XIPH_PATH_OGG,
[dnl 
AC_ARG_VAR([OGG_PREFIX],[path to ogg installation])
AC_ARG_WITH(ogg,
    [AC_HELP_STRING([--with-ogg=PREFIX],
                   [Prefix where libogg is installed (optional)])],
    ogg_prefix="$withval",
    ogg_prefix="$OGG_PREFIX"
)
if test "x$ogg_prefix" = "x"; then
    if test "x$prefix" = "xNONE"; then
        ogg_prefix=/usr/local
    else
        ogg_prefix="$prefix"
    fi
fi

XIPH_GCC_WARNING([-I$ogg_prefix/include],,
        [OGG_CFLAGS="-I$ogg_prefix/include"
        OGG_LDFLAGS="-L$ogg_prefix/lib"
        ])
OGG_LIBS="-logg"

#
# check if the installed Ogg is sufficiently new.
#
AC_MSG_CHECKING([for ogg_sync_init in libogg])
ac_save_CFLAGS="$CFLAGS"
ac_save_LIBS="$LIBS"
ac_save_LDFLAGS="$LDFLAGS"
CFLAGS="$CFLAGS $OGG_CFLAGS"
LIBS="$LIBS $OGG_LIBS"
LDFLAGS="$LDFLAGS $OGG_LDFLAGS"
AC_TRY_LINK_FUNC(ogg_sync_init,
        [ifelse([$1],, [AC_MSG_RESULT([ok])], [$1])],
        [AC_TRY_LINK([#include <ogg/ogg.h>],, 
            [ ifelse([$2], ,[AC_MSG_ERROR([found, but needs updating])], [$2])],
            [ ifelse([$2], ,[AC_MSG_ERROR([not found, maybe you need to set LD_LIBRARY_PATH or /etc/ld.so.conf])], [$2])])
        ])
CFLAGS="$ac_save_CFLAGS"
LDFLAGS="$ac_save_LDFLAGS"
LIBS="$ac_save_LIBS"

AC_SUBST(OGG_CFLAGS)
AC_SUBST(OGG_LDFLAGS)
AC_SUBST(OGG_LIBS)
])

{'fraction_non_alphanumeric': 0.15902045877247364, 'fraction_numerical': 0.012399256044637322, 'mean_word_length': 4.3604651162790695, 'pattern_counts': {'":': 0, '<': 17, '<?xml version=': 0, '>': 17, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2159721', 'n_tokens_mistral': 1355, 'n_tokens_neox': 1301, 'n_words': 240}